


Cylindrical Coordinates

```
> restart:with(plots):with(Student):with(MultivariateCalculus):  
> setoptions3d(axes=NORMAL,labels=["x","y","z"],orientation=[20,  
70]);
```

Cylindrical Coordinates.


We use the command `coordplot3d` from the `plots` package to give us a graphical representation of the cylindrical coordinates $[r$ (red), θ (green), z (blue)].

```
> coordplot3d(cylindrical);
```




Let's draw a cylinder using `coords=cylindrical`. The first argument is r as a function of θ and z .

```
> plot3d(1/2,theta=0..2*Pi,z=-2..2,coords=cylindrical,style=  
patchnogrid);
```




We plot a cone.

```
> plot3d(z, theta=0..2*Pi, z=0..1, coords=cylindrical, style=patchnogrid);
```



We can also do parametrized cylindrical plots with $[r, \theta, z]$ as functions of the parameters s and t , for instance .

```
> plot3d([s*t, s, cos(t^2)], s=0..Pi, t=-2..2, coords=cylindrical,  
style=patchnogrid);
```



Next we do a cylindrical triple integral.

```
> cylinder_integral:=Int(Int(Int(r^3-2*theta,z=r^2..36-r^2),r=0.
.3),theta=0..2*Pi)=int(int(int(r^3-2*theta,z=r^2..36-r^2),r=0.
.3),theta=0..2*Pi);
```

$$cylinder_integral := \int_0^{2\pi} \int_0^3 \int_{r^2}^{36-r^2} (r^3 - 2\theta) \, dz \, dr \, d\theta = 972\pi - 360\pi^2$$

>


Suppose we wish to find the volume of a cone, 90 degrees at the vertex, topped by a sphere of radius 1.

```
> with(plottools);
```

```
[arc, arrow, circle, cone, cuboid, curve, cutin, cutout, cylinder, disk, dodecahedron, ellipse, ellipticArc,
hemisphere, hexahedron, homothety, hyperbola, icosahedron, line, octahedron, parallelepiped,
pieslice, point, polygon, project, rectangle, reflect, rotate, scale, semitorus, sphere, stellate,
tetrahedron, torus, transform, translate, vml]
```

```
> p1:=plot3d(z,theta=0..2*Pi,z=0..1/sqrt(2),coords=cylindrical,
style=patchngrid):
p2:=plot3d(sqrt(1-z^2),theta=0..2*Pi,z=1/sqrt(2)..1,coords=
cylindrical,style=patchngrid):
p3:=plot3d([1/sqrt(2),s,1/sqrt(2)],s=0..2*Pi,t=-2..2,coords=
cylindrical):
```

```
display(p1,p2,p3);
```



We first find the volume using cylindrical coordinates.

```
> cylinder_volume:=Int(Int(Int(r,z=r..sqrt(1-r^2)),r=0..sqrt(1/2)),theta=0..2*Pi)=int(int(int(r,z=r..sqrt(1-r^2)),r=0..sqrt(1/2)),theta=0..2*Pi);
```

$$cylinder_volume := \int_0^{2\pi} \int_0^{\frac{1}{2}\sqrt{2}} \int_r^{\sqrt{1-r^2}} r \, dz \, dr \, d\theta = \frac{1}{2} \pi^{3/2} \left(\frac{4}{3\sqrt{\pi}} - \frac{1}{3} \frac{\sqrt{2}}{\sqrt{\pi}} \right) - \frac{1}{6} \sqrt{2} \pi$$

```
> simplify(%);
```

$$\left(\int_0^{\frac{1}{2}\sqrt{2}} r \left(\int_r^{\sqrt{1-r^2}} 1 \, dz \right) dr \right) \left(\int_0^{2\pi} 1 \, d\theta \right) = \frac{2}{3} \pi - \frac{1}{3} \sqrt{2} \pi$$

Finally, we find the volume using Cartesian coordinates.

```
> Cartesian_volume:=Int(Int(Int(1,z=sqrt(x^2+y^2)..sqrt(1-x^2-y^2)),y=-sqrt(1/2-x^2)..sqrt(1/2-x^2)),x=-sqrt(1/2)..sqrt(1/2))=int(int(int(1,z=sqrt(x^2+y^2)..sqrt(1-x^2-y^2)),y=-sqrt(1/2-x^2)..sqrt(1/2-x^2)),x=-sqrt(1/2)..sqrt(1/2));
```

It took Brother Walter's computer 404.26 seconds to determine that Maple could only do the first of the three integrals. Let's try to do the integral numerically.

```
> Cartesian_volume:=Int(Int(Int(1,z=sqrt(x^2+y^2)..sqrt(1-x^2-y^2)
),y=-sqrt(1/2-x^2)..sqrt(1/2-x^2)),x=-sqrt(1/2)..sqrt(1/2))
=evalf(Int(Int(Int(1,z=sqrt(x^2+y^2)..sqrt(1-x^2-y^2)),y=-sqrt
(1/2-x^2)..sqrt(1/2-x^2)),x=-sqrt(1/2)..sqrt(1/2)));
```

It took brother Walter's computer 2705.34 seconds (over 45 minutes) to do this approximation.

```
>
```